

The ultimate in-depth reference

Hundreds of timesaving solutions

Supremely well-organized, packed  
with expert advice



# Microsoft Excel Inside **OUT**

(Office 2021 and Microsoft 365)

**Bill Jelen**

# Microsoft Excel Inside Out

## (Office 2021 and Microsoft 365)

**Bill Jelen**

*MrExcel*

The examples in the following sections reveal details on how to use the lookup functions and how to combine them to create powerful results.

## Using the CHOOSE function for simple lookups

Most lookup functions require you to set up a lookup table in a range on the worksheet. However, the CHOOSE function enables you to specify up to 254 choices right in the syntax of the function. The formula that requires the lookup should be able to calculate an integer from 1 to 254 to use the CHOOSE function.

### Syntax:

`CHOOSE(index_num,value1,value2,...)`

The CHOOSE function chooses a value from a list of values, based on an index number. The CHOOSE function takes the following arguments:

- **index\_num**: This specifies which value argument is selected. **index\_num** must be a number between 1 and 254 or a formula or reference to a cell containing a number between 1 and 254:
  - If **index\_num** is 1, CHOOSE returns **value1**; if it is 2, CHOOSE returns **value2**; and so on.
  - If **index\_num** is a decimal, it is rounded down to the next lowest integer before being used.
  - If **index\_num** is less than 1 or greater than the number of the last value in the list, CHOOSE returns a #VALUE! error.
- **value1,value2,...**: These are 1 to 254 value arguments from which CHOOSE selects a value or an action to perform based on **index\_num**. The arguments can be numbers, cell references, defined names, formulas, functions, or text.

The example in Figure 9.9 shows survey data from some respondents. Columns B:F indicate their responses on five measures of your service. Column G calculates an average that ranges from 1 to 5. Suppose that you want to add words to column H to characterize the overall rating from the respondent. The following formula is used in cell H4:

`=CHOOSE(G4,"Strongly Disagree","Disagree","Neutral","Agree","Strongly Agree")`

	A	B	C	D	E	F	G	H	I	J	K
1	<b>Survey Results</b>										
2											
3	<b>Resp.</b>	<b>Q1</b>	<b>Q2</b>	<b>Q3</b>	<b>Q4</b>	<b>Q5</b>	<b>Avg.</b>	<b>Result</b>			
4	R01	5	5	5	5	4	4.8	Agree			
5	R02	4	4	4	4	4	4	Agree			
6	R03	1	2	2	2	2	1.8	Strongly Disagree			
7	R04	5	4	4	3	4	4	Agree			
8	R05	3	3	2	2	2	2.4	Disagree			

Figure 9.9 CHOOSE is great for simple choices in which the index number is between 1 and 254.

## Moving from VLOOKUP to XLOOKUP

In 2019, the XLOOKUP function was introduced in Excel. It is designed to replace VLOOKUP, HLOOKUP, and INDEX/MATCH. The Excel Calc team, lead by Joe McDaid, wanted to remove many of the potential errors that happen with the old VLOOKUP and HLOOKUP scenarios.

Although XLOOKUP offers a total of six arguments, only the first three are required to perform an exact match XLOOKUP. To build an exact match XLOOKUP that returns a single value, you will specify:

- **Lookup\_value:** This is exactly like you would use in HLOOKUP or VLOOKUP.
- **Lookup\_Array:** This is a single column or row where the **Lookup\_value** should be found. Typically with a VLOOKUP, you would specify a rectangular range. When using XLOOKUP, you will specify only the first column or row of the range.
- **Return\_array:** This is a range of the same shape as a **Lookup\_Array**. If the **Lookup\_value** is found in the fifth position of the **Lookup\_Array**, then XLOOKUP will return the fifth item from the **return\_array**.

Consider the data in Figure 9.10. To find the sales rep name associated with the ID in A2 using VLOOKUP, you would have used this formula:

```
=VLOOKUP(A2,$H$2:$I$7,2,FALSE)
```

To perform the same task with XLOOKUP, you would use:

```
=XLOOKUP(A2,$H$2:$H$7,$I$2:$I$7)
```

Both formulas start by pointing to cell A2 as the **lookup\_value**. While VLOOKUP points to a 2-D range of H2:I7, XLOOKUP only points to the column containing the Rep ID numbers H2:H7. While VLOOKUP specifies that the function should return the second column from **lookup\_table**, the XLOOKUP function instead points to a **return\_array** of I2:I7.

For exact match scenarios, VLOOKUP would end in False or zero. This is not required with XLOOKUP because XLOOKUP defaults to an exact match.

	A	B	C	D	E	F	G	H	I	J
1	Rep	Date	Sale Amt	Rep Name	Office		Office	Rep	Name	
2	R5	2/17/2025	168.02	=XLOOKUP(A2,\$H\$2:\$H\$7,\$I\$2:\$I\$7)			Chicago	R4	Amar	
3	R7	2/17/2025	130.56	Michael			Springfield	R8	Jerry	
4	R5	2/17/2025	124.48	Manny			Kansas City	R6	Linda	
5	R3	2/17/2025	128.63	Marc			Tulsa	R5	Manny	
6	R6	2/19/2044	116.79	Linda			Dallas	R3	Marc	
7	R3	2/19/2044	113.92	Marc			Austin	R7	Michael	
8	R8	2/19/2044	113.85	Jerry						
9	R8	2/19/2044	156.19	Jerry						
10	R2	2/20/2063	103.45	#N/A						
11	R3	2/20/2063	186.49	Marc						
12	R4	2/20/2063	132.9	Amar						

**Figure 9.10** XLOOKUP accomplishes in three arguments what would have required four arguments with VLOOKUP.

The preceding example illustrates the first three benefits of XLOOKUP over VLOOKUP:

- XLOOKUP assumes that you are doing an exact match. Most VLOOKUPs were performing exact matches. Because VLOOKUP defaulted to a range match, anyone who left off the fourth argument of VLOOKUP could potentially get the wrong results.
- XLOOKUP uses a range reference for the return\_array instead of VLOOKUP using an integer-based column index number. Imagine if a coworker inserted a column between columns H and I in Figure 9.10. The VLOOKUP would break. The XLOOKUP will continue to point to the correct range.
- XLOOKUP would have no problem returning a result that is found to the left of the lookup\_value. If you need to return the office from column G to cell E2, the XLOOKUP formula would be =XLOOKUP(A2, \$H\$2:\$H\$7, \$G\$2:\$G\$7). This would have been extremely complicated with VLOOKUP.

### Using the optional arguments in XLOOKUP

The function syntax for XLOOKUP offers three optional arguments:

=XLOOKUP([lookup\_value],Lookup\_Array,return\_array,[if\_not\_found],[match\_mode],[search\_mode])

Notice that the XLOOKUP in cell D10 is returning an #N/A error because sales rep R2 is not found in the lookup table. To avoid this error with VLOOKUP, you would wrap the VLOOKUP function in the IFNA or IFERROR function. In contrast, XLOOKUP has the If\_Not\_Found argument as

an optional fourth argument. Simply specify =XLOOKUP(A2,\$H\$2:\$H\$7,\$I\$2:\$I\$7,"Not Found").

The match\_mode optional argument offers four choices:

- 0—Exact match. This is the default.
- 1—Exact match or next larger item. In contrast to the old MATCH function, the lookup\_range does not have to be sorted to use this option.
- -1—Exact match or next smaller item. The lookup range does not have to be sorted for this option.
- 2—Wildcard search. Many people did not realize that VLOOKUP, HLOOKUP, and MATCH treated ?, \*, and ~ as wildcards. By default, XLOOKUP will treat ?, \*, and ~ as text. If you specifically need to have them treated as wildcards, then specify 2 for the match\_mode.

Figure 9.11 shows XLOOKUP using ranges for a grading scale. If a student scores a 92, XLOOKUP will return an A because the match\_mode argument is looking for the row equal to or larger than 92.

	A	B	C	D	E	F	G	H
1	Student	Score	Grade					
2	CECILIA ALBERT	100	A		A	100		
3	KRISTEN FOREMAN	90	A		B	89		
4	DOROTHY MARTINEZ	89	B		C	79		
5	LUIS FISCHER	80	B		D	69		
6	TERRY SKINNER	79	C		F	65		
7	SHELLEY OSBORN	70	C					
8	MARVIN SCHULTZ	69	D					
9	JOE TRUJILLO	65	F					
10	WILLIAM SKINNER	64	F					
11	CAROLYN CURRY	68	D					
12	CARL ZAMORA	96	A					

**Figure 9.11** The XLOOKUP formula in column C finds the correct grade from the table in columns E and F.

The optional sixth argument for XLOOKUP is the search\_mode. It accepts these values:

- 1—Search first to last. This is how VLOOKUP and MATCH work. It is the default for XLOOKUP.
- -1—Search last to first allows you to search from the bottom of the list. This provides a new feature that MATCH and VLOOKUP could not offer, since they always search from the top of the list.

- 2—Binary search, from the top of the list. Before 2018, binary searches were faster than regular searches. The Excel team rewrote the search logic used in VLOOKUP, MATCH, and HLOOKUP. There is no longer any advantage to using a binary search. There is one disadvantage: The binary search option requires the lookup\_range to be sorted ascending.
- -2—Binary search, from the bottom of the list. This option requires the lookup\_range to be sorted in descending sequence. The Excel team does not expect anyone to use the binary search options because they are not faster than using 1 or -1.

Using one XLOOKUP to return multiple values

Thanks to the new Dynamic Arrays, XLOOKUP can return multiple answers. There are two different uses for this feature.

In Figure 9.12, the formula sends 64 cells in to the lookup\_value argument. The formula returns a match for each of the 64 values. A single formula of =XLOOKUP(A2:A65,\$H\$2:\$H\$7,\$G\$2:\$G\$7,"Not Found") in cell E2 returns the Office column in E2:E65.

E2	=XLOOKUP(A2:A65,\$H\$2:\$H\$7,\$G\$2:\$G\$7,"Not Found")								
	A	B	C	D	E	F	G	H	I
1	Rep	Date	Sale Amt	Rep Name	Office		Office	Rep	Name
2	R5	2/17/2025	168.02	Manny	Tulsa		Chicago	R4	Amar
3	R7	2/17/2025	130.56	Michael	Austin		Springfield	R8	Jerry
4	R5	2/17/2025	124.48	Manny	Tulsa		Kansas City	R6	Linda
5	R3	2/17/2025	128.63	Marc	Dallas		Tulsa	R5	Manny
6	R6	2/19/2044	116.79	Linda	Kansas City		Dallas	R3	Marc
7	R3	2/19/2044	113.92	Marc	Dallas		Austin	R7	Michael
8	R8	2/19/2044	113.85	Jerry	Springfield				
9	R8	2/19/2044	156.19	Jerry	Springfield				
10	R2	2/20/2063	103.45	#N/A	Not Found				
11	R3	2/20/2063	186.49	Marc	Dallas				

Figure 9.12 A single XLOOKUP formula in E2 returns all of the values in the Office column.

Another use for returning multiple answers is when your lookup table contains several columns of answers. In Figure 9.13, the table contains a value for each month. To return data from all 12 months, you would specify a return\_array that is 12 columns wide.

In contrast to using VLOOKUP or INDEX/MATCH, the formula shown in Figure 9.13 can find the lookup\_value once for each item instead of having to find the lookup\_value 12 times for each item. This means XLOOKUP is faster than VLOOKUP.



Figure 9.13 shows a single XLOOKUP formula in cell C5. The formula bar displays `=XLOOKUP(B5,$A$17:$A$14056,$B$17:$M$14056)`. The spreadsheet contains a table with 12 columns (Jan-Dec) and 12 rows of data. Cell C5 contains the value 0, which is the result of the XLOOKUP formula.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1	One XLOOKUP returns all 12 columns														
2	<code>=XLOOKUP(B5,\$A\$17:\$A\$14056,\$B\$17:\$M\$14056)</code>														
3															
4	Qty	Item	Jan	Feb	Mar	Apr	May	Jun	Jul	Aug	Sep	Oct	Nov	Dec	
5	2	C529	0	5	0	81	96	5	5	0	2	0	4	2	
6	10	F708	1	2	0	88	1	0	53	2	0	0	80	47	
7	9	X291	0	80	0	0	0	50	29	33	0	87	1	4	
8	1	E890	2	87	0	1	2	0	5	0	95	4	78	2	
9	5	C299	0	2	4	0	48	0	2	0	89	66	0	3	
10	4	S123	4	5	3	69	5	0	4	102	97	0	30	4	
11	1	V600	131	48	129	1	0	1	0	105	67	117	105	4	
12	9	P765	3	0	70	2	1	3	0	0	0	0	0	91	
13															
14	Inventory By Month														
15															
16	Item		Jan	Feb	Mar	Apr	May	Jun	Jul	Aug	Sep	Oct	Nov	Dec	
17	G245		36	2	71	1	96	83	0	34	81	59	0	0	
18	J535		0	3	0	0	0	85	5	137	0	32	0	5	
19	X286		113	2	2	0	0	35	29	0	58	4	0	1	
20	M562		4	72	0	58	5	0	110	133	0	3	0	136	
21	V241		1	120	0	4	105	0	1	134	0	0	0	3	
22	I238		1	0	48	0	0	5	131	40	0	132	2	0	
23	E599		0	24	122	4	3	4	30	75	4	5	0	0	
24	B372		0	80	0	1	100	0	0	1	4	0	5	0	
25	M769		2	85	0	141	0	4	33	3	0	0	0	112	
26	E187		0	56	0	131	115	0	2	1	4	130	2	0	
27	I664		5	0	4	0	125	3	3	40	67	4	0	89	
28	F142		0	5	5	83	5	0	0	71	90	0	0	3	

Figure 9.13 A single XLOOKUP formula in C5 returns all 12 months of answers.

The techniques shown in Figures 9.12 and 9.13 are nice improvements over VLOOKUP. The natural next step is to combine the two techniques to look up all rows and columns in a single formula. As of June 2021, this is not supported by Excel. In Figure 9.13, a formula in C5 of `=XLOOKUP(B5:B12,A17:A14056,B17:M14056)` only returns one column of answers. Excel can extend the array in one direction or the other but not in both directions at the same time.

### Using XLOOKUP to search sideways

The XLOOKUP function can also replace the old HLOOKUP and LOOKUP functions. HLOOKUP was used when your lookup table was arranged horizontally. In Figure 9.14, the `Lookup_Array` is B5:F5. The `return_array` is B6:F6.

Figure 9.14 shows a sideways XLOOKUP formula in cell B2. The formula bar displays `=XLOOKUP(B1,B5:F5,B6:F6)`. The spreadsheet contains a table with 6 columns (Name, Dept, Andy, Barb, Chris, Diane, Ed) and 6 rows of data. Cell B2 contains the value 'Sales', which is the result of the XLOOKUP formula.

	A	B	C	D	E	F
1		Chris				
2		Sales				
3						
4						
5	Name	Andy	Barb	Chris	Diane	Ed
6	Dept	Accounting	Marketing	Sales	Engineering	Human Resources
7						
8						
9						

Figure 9.14 Search for Chris in row 5 and return the department from row 6.

The LOOKUP function is one of the earliest lookup functions. It would allow you to do all of the look-ups in one formula. It also had the unusual ability to search a vertical array and return the answer